Conjectures, Statements and Proofs

„I want to know how God created the world."
(Albert Einstein)

Einstein wanted to know how God had created the world and he did all his best to answer this question. However, we have more modest aims: we want to know if a generic target computer-algebra system can support the drawing up of the conjectures and their proofs concerning the solutions of the tasks and if yes, how. We keep on experimenting with Maple while we are going to meet new features of the mathematical problem solving.

In the beginning, Maple was used for illustration: it helped to understand the tasks to be solved and the solutions received. Then we used it as a tool for gaining some experience. With the help of the experience gained we drew up conjectures and then we proved them. Sometimes, especially in the case of the proofs, working with a computer-algebra system proved to be useful. However, these operations were done during special circumstances by creating an environment in which nothing can disturb the observation of the phenomena in question.

However, we are going to solve two complex mathematical problems in this chapter by using all the computer-algebra armouries mentioned. The technical worksheet of the chapter sheds light on the most exciting yet most debated field of the computer-algebra systems, on the problem of the automatic simplification of the expressions and the manipulation of the formulas.

The Chebyshev Polynomials

Consider the following definition of the Chebyshev polynomials:

$$T_k := \cos(k \arccos(x))\text{ , ahol } k = 0, 1, 2...$$

Draw up our conjectures for the degree of the kth Chebyshev polynomial and for the coefficient of the highest degree term.

Wait a minute: what do we mean by polynomial and highest degree term? For the time being, we can only see trigonometric functions so why are the polynomials mentioned? First, let's convince ourselves that the task is correct.

The following for cycle creates the ith Chebyshev polynomial for the i=1,2,...8 indexes in the Chebyshevi variable. The subsequent commands can be used for experimenting. We give a value for the variable m then we display the definition of the mth Chebyshev polynomial and explain it in two steps. In the first step, we do it by freezing the arcos(x) then in the second step we do it without that.

```
> n:=8:
  for i from 0 to n do
   Csebisev[i]:=cos(i*arccos(x)):
  od:
```

> $m := 2$

$$2 \qquad (1)$$

> $Csebisev_m$

$$\cos(2 \arccos(x)) \qquad (2)$$

> $expand(\%, \arccos(x))$

(3)

$$2\cos(\arccos(x))^2 - 1 \quad (3)$$

> *expand*(%)

$$2x^2 - 1 \quad (4)$$

We encourage you to experiment further. Step back with the cursor and change 2 to 3 or 5 or as you like it in the m:=2 instruction. Since the Chebyshev8 is the largest Chebyshev polynomial calculated, we get an error message for the values bigger than 8. Don't be afraid of it: the system gives correct outputs for the values below 8.

So what is going on here exactly? The $\cos(k\arccos(x))$ is really a polynomial of the $\cos(\arccos(x))$ and concerning the $\cos(\arccos(x)) = x$ identity in this case we really get the polynomial of the x.

Now that we have persuaded ourselves that we were right about the initial problem, it would be good to see the polynomials. We are going to show only one graph out of the nine image animation sequences. But our readers should feel free to experiment for a longer period in the animation window. The graphs are nice and it is worth spending time on them

```
> for i from 0 to n do
  p||i:=plot(Csebisev[i], x=-1..1, scaling=constrained,color=red);
  x||i:=plots[textplot]([0.7, 0.8, convert(expand(Csebisev[i]),
  string)],colour=blue,align=LEFT);
  d||i:=plots[display]([p||i,x||i]):
  od:
  plots[display]([d||(0..n)], insequence=true, title='`Csebisev
  polinomok a (-1,1) intervallumon`');
```



Let's look at the coefficients of the terms with the highest exponent of the Chebyshev polynomials.

```
> for i from 0 to n  do
  printf(`%s%d%4d  %a\n`,'`n=`', i, lcoeff(expand(Csebisev[i],x)),
  expand(Csebisev[i]));
  od:
n=0   1  1
n=1   1  x
n=2   2  2*x^2-1
n=3   4  4*x^3-3*x
n=4   8  8*x^4-8*x^2+1
n=5  16  16*x^5-20*x^3+5*x
n=6  32  32*x^6-48*x^4+18*x^2-1
n=7  64  64*x^7-112*x^5+56*x^3-7*x
n=8 128  128*x^8-256*x^6+160*x^4-32*x^2+1
```

Remember that the lcoeff procedure returns the coefficient of the highest degree term of the polynomial given as a parameter. The printf procedure, as the tool of the formatted display, is known for those who are familiar with the C language. The Maple procedures known so far produce 2-D, centred, graphical outputs. Notice, however, that the printf puts its output on the left side and represents the powers not as a denominator but by the usage of the ^ sign within a line, characterically. Although this screen does not look so nice we have chosen it because it is more suitable for the table-like layout.

It is difficult to draw up our conjecture based on the output above. For every k natural number

- the degree of the kth Chebyshev polynomial is k
- and the coefficient of the xk i 2k-1.

Let's look at the first proof. We should also use the whole induction in this case. The first step of the induction is obvious. The first Chebyshev polynomial is x the degree of which is 1 and its coefficient is 20. It is correct so far. Let's assume that our statement is already true for the k and consider the k+1st Chebyshev polynomial.

> cos((k + 1)* arccos(x))

$$\cos((k+1)\arccos(x)) \quad \textbf{(5)}$$

> *expand*(%)

$$\cos(\arccos(x)\,k)\,x - \sin(\arccos(x)\,k)\,\sqrt{1-x^2} \quad \textbf{(6)}$$

The result of the expand procedure is satisfying because the $\cos(k\arccos(x))$ appears about which we can make a statement based on the induction conjecture. But at the same time the $\sin(k\arccos(x))$ also appears about which we do not know anything despite such a long preparation. By all means, let's look at the behaviour of the latter expression for the k+1.

> sin((k + 1)* arccos(x))

$$\sin((k+1)\arccos(x)) \quad \textbf{(7)}$$

> *expand*(%)

$$\sin(\arccos(x)\,k)\,x + \cos(\arccos(x)\,k)\,\sqrt{1-x^2} \quad \textbf{(8)}$$

Well, the $\cos(k\arccos(x))$ has appeared. Maybe we have reached a dead-end. But if we give it some thought, we can understand that the problem is that we should be able to make statements about both the $\cos(k\arccos(x))$ and the $(k\arccos(x))$ at the induction step. To do this, we should prove the properties of the $\cos(k\arccos(x))$ and $\sin(k\arccos(x))$ within a statement. But what should this statement be? Again, let's use Maple to gain some experience.

> u := 2 :

> *expand*(sin(u * arccos(x)))

$$2\sqrt{1-x^2}\,x \quad \textbf{(9)}$$

Maybe you are aware of what is going to happen and you might be fed up with it. But still: keep on experimenting, gaining some experience and drawing up the properties of the mathematical object based on experience. Here we take an oath that you will not have to cope with so much difficulty in the rest of the book.

In the cycle to be executed below, we are going to use a typical Maple method with the freezing of the sub-expressions. Since we know from the experience gained previously (supposing you did it) that the $\sin(u\arccos(x))$expression appears in the $\sqrt{1-x^2}$ expression. We will substitute this with the variable z and then we create a contracted form of the polynomial received by the z with the collect procedure and finally we substitute the original expression to the place of the z.

```
> m:=6:
  for u from 1 to m do
   expand(sin(u*arccos(x))):
   subs(sqrt(1-x^2)=z,%); collect(%,z); subs(z=sqrt(1-x^2),%);
   print(u,%, expand(cos(u*arccos(x)))):
  od:
```

$$1, \sqrt{1-x^2}, x$$

$$2, 2\sqrt{1-x^2}\, x, 2x^2-1$$

$$3, (4x^2-1)\sqrt{1-x^2}, 4x^3-3x$$

$$4, (8x^3-4x)\sqrt{1-x^2}, 8x^4-8x^2+1$$

$$5, (16x^4-12x^2+1)\sqrt{1-x^2}, 16x^5-20x^3+5x$$

$$6, (32x^5-32x^3+6x)\sqrt{1-x^2}, 32x^6-48x^4+18x^2-1 \tag{10}$$

Our statement (or conjecture) may be the following: for all 1<k integers

- $\cos(k \arccos(x)) = 2^{(k-1)} \cdot x^k + P$
- $\sin(k \arccos(x)) = \left(2^{(k-1)} \cdot x^{(k-1)} + Q\right) \cdot \sqrt{1-x^2}$,

in which case P is k-2nd degree and Q is a k-3rd degree polynomial or constant at best.

To execute the whole induction proof, we entered the 1. statement and 2. statement variables with the general syntax of the two statements to be proved and we executed the first step of the proof.

> $1.\ \acute{a}ll\acute{i}t\acute{a}s := \cos(\arccos(x)\,k) = 2^{(k-1)}\, x^k + P$

$$\cos(\arccos(x)\,k) = 2^{(k-1)}\, x^k + P \tag{11}$$

> $2.\ \acute{a}ll\acute{i}t\acute{a}s := \sin(\arccos(x)\,k) = \left(2^{(k-1)}\, x^{(k-1)} + Q\right)\sqrt{1-x^2}$

$$\sin(\arccos(x)\,k) = \left(2^{(k-1)}\, x^{(k-1)} + Q\right)\sqrt{1-x^2} \tag{12}$$

It can be easily seen that the statements are fulfilled in the case of k=1 and by choosing P=Q=0.

Our induction conjecture is that both the 1. and the 2. statement equalities are fulfilled for the P and Q polynomials chosen correctly. Let’s look at the left side of the 1. statement for k+1 instead of the k.

> $expand(subs(k=k+1, lhs(1.\ \acute{a}ll\acute{i}t\acute{a}s)))$

$$\cos(\arccos(x)\,k)\,x - \sin(\arccos(x)\,k)\sqrt{1-x^2} \tag{13}$$

We can execute the induction step by substituting the left sides of the 1. and 2. statement equalities with their right sides in the result of the previous operation.

> $subs(lhs(1.\ \acute{a}ll\acute{i}t\acute{a}s) = rhs(1.\ \acute{a}ll\acute{i}t\acute{a}s), lhs(2.\ \acute{a}ll\acute{i}t\acute{a}s) = rhs(2.\ \acute{a}ll\acute{i}t\acute{a}s), \%)$

$$\left(2^{(k-1)}\, x^k + P\right)x - \left(2^{(k-1)}\, x^{(k-1)} + Q\right)(1-x^2) \tag{14}$$

> $simplify(\%)$

$$x^{(k+1)}\, 2^k + x\,P - 2^{(k-1)}\, x^{(k-1)} - Q + Q\,x^2 \tag{15}$$

Because the degree of P is k-2 at best that's why the degree of the x.P is k-1 at best. Similarly, the degree of the Q is k-2 at best that's why the degree of the $Q.x^2$ is also k-1 at best. According to this, the syntax is $2^k \cdot x^{(k+1)} + R$ for the k+1st degree Chebyshev polynomial in which case the degree of the R is k-1 at best. The degree of this polynomial is k+1 and the coefficient of the term with the highest degree is 2^k. We have finished the first part of the induction step.

Let's look at the statement concerning the $\sin(k \arccos(x))$ for k+1 instead of the k.

> $expand(subs(k=k+1, lhs(2.\ \text{állítás})))$

$$\sin(\arccos(x)\,k)\,x + \cos(\arccos(x)\,k)\,\sqrt{1-x^2} \quad \textbf{(16)}$$

> $subs(lhs(1.\ \text{állítás}) = rhs(1.\ \text{állítás}), lhs(2.\ \text{állítás}) = rhs(2.\ \text{állítás}), \%)$

$$\left(2^{(k-1)}\,x^{(k-1)} + Q\right)\sqrt{1-x^2}\,x + \left(2^{(k-1)}\,x^k + P\right)\sqrt{1-x^2} \quad \textbf{(17)}$$

> $subs\left(\sqrt{1-x^2} = z, \%\right)$

$$\left(2^{(k-1)}\,x^{(k-1)} + Q\right)z\,x + \left(2^{(k-1)}\,x^k + P\right)z \quad \textbf{(18)}$$

> $subs\left(z = \sqrt{1-x^2}, \%\right)$

$$\left(2^{(k-1)}\,x^{(k-1)} + Q\right)\sqrt{1-x^2}\,x + \left(2^{(k-1)}\,x^k + P\right)\sqrt{1-x^2} \quad \textbf{(19)}$$

We let you finish the induction proof.

Let's continue the examination of the Chebyshev polynomials and look for a recursive formula for them. Naturally, we start with gaining some experience. We are going to look for relations between the kth degree polynomial and the smaller index polynomials. Let's enter the k-1st and the k-2nd Chebyshev polynomials

> $Csebisev_{k-1} := expand(\cos((k-1)\arccos(x)))$

$$\cos(\arccos(x)\,k)\,x + \sin(\arccos(x)\,k)\,\sqrt{1-x^2} \quad \textbf{(20)}$$

> $Csebisev_{k-2} := expand(\cos((k-2)\arccos(x)))$

$$2\cos(\arccos(x)\,k)\,x^2 - \cos(\arccos(x)\,k) + 2\sin(\arccos(x)\,k)\,\sqrt{1-x^2}\,x \quad \textbf{(21)}$$

>

At first sight, it seems hopeless but don't give it up! Let's look at the terms with roots. Their only difference lies in the 2.x multiplier. If we extract the Chebyshev(k-2) polynomial from the 2.x fold of the Chebyshev(k-1) then the terms with roots must fall out

> $2\,x\,Csebisev_{k-1} - Csebisev_{k-2}$

$$2\,x\left(\cos(\arccos(x)\,k)\,x + \sin(\arccos(x)\,k)\,\sqrt{1-x^2}\right) - 2\cos(\arccos(x)\,k)\,x^2 + \cos(\arccos(x)\,k) - 2\sin(\arccos(x)\,k)\,\sqrt{1-x^2}\,x \quad \textbf{(22)}$$

> $expand(\%)$

$$\cos(\arccos(x)\,k) \quad \textbf{(23)}$$

```
>
```

Great! We have found the solution. We can see the kth Chebyshev polynomial. According to this, the Chebyshev polynomials fulfil the following recursive formula.

$$T_0 = 1,\ T_1 = x,\ T_k = 2 \cdot x \cdot T_{k-1} - T_{k-2}$$

$$(1 < k).$$

After this we can easily enter the cycle which calculates the nth Chebyshev polynomial for an arbitrary n by using the recursive formula.

```
> n:=12: # alkalom a kisérletezésre
  if n=0 then `aktuális`:= 1
  elif n=1 then `aktuális`:= x
  else
   `kettvel elbbi`:=1: `elz`:=x:
   for i from 2 to n do
    `aktuális`:=expand(2*x*`elz`-`kettvel elbbi`);
    `kettvel elbbi`:=`elz`;
    `elz`:=`aktuális`;
   od:
  fi:
  `aktuális`;
```

$$12$$

$$2048\,x^{12} - 6144\,x^{10} + 6912\,x^{8} - 3584\,x^{6} + 840\,x^{4} - 72\,x^{2} + 1 \quad \textbf{(24)}$$

We have become familiar with and proved several properties of the Chebyshev polynomials. We have only one question left: is there a closed formula for the Chebyshev polynomials? The system proves to be really beneficial at this point because, as we have seen, the Chebyshev polynomials fulfil the recurrent relation which the rsolve procedure easily solves

> $\{C(k) = 2\,x\,C(k-1) - C(k-2),\ C(0) = 1,\ C(1) = x\}$

$$\{C(k) = 2\,x\,C(k-1) - C(k-2),\ C(0) = 1,\ C(1) = x\} \quad \textbf{(25)}$$

> $rsolve(\{C(k) = 2\,x\,C(k-1) - C(k-2),\ C(0) = 1,\ C(1) = x\},\ C(\,))$

$$\frac{1}{2}\,\frac{\left(1 - x^2 - x\sqrt{x^2-1}\right)\left(-\frac{1}{-x-\sqrt{x^2-1}}\right)^k}{\sqrt{x^2-1}\left(-x-\sqrt{x^2-1}\right)} + \frac{1}{2}\,\frac{\left(-1 + x^2 - x\sqrt{x^2-1}\right)\left(-\frac{1}{-x+\sqrt{x^2-1}}\right)^k}{\sqrt{x^2-1}\left(-x+\sqrt{x^2-1}\right)} \quad \textbf{(26)}$$

> $normal(\%, expanded)$

$$\frac{1}{2}\left(\frac{1}{x+\sqrt{x^2-1}}\right)^k + \frac{1}{2}\left(-\frac{1}{-x+\sqrt{x^2-1}}\right)^k \quad \textbf{(27)}$$

We can see the kth Chebyshev polynomial given by a closed algebraic formula. It is questionable if this formula really gives the Chebyshev polynomials. In order to develop your trust in Maple, we are going to show that the formula fulfils the recursive relations received for the Chebyshev polynomials. Of course, we check it with Maple.

> $subs(k=0, \textbf{(27)})$

$$1 \qquad \textbf{(28)}$$

> $subs(k=1, \textbf{(27)})$

$$\frac{1}{2}\frac{1}{x+\sqrt{x^2-1}}-\frac{1}{2}\frac{1}{-x+\sqrt{x^2-1}} \qquad \textbf{(29)}$$

> $normal(\%, expanded)$

$$x \qquad \textbf{(30)}$$

> $\textbf{(27)} - 2\, subs(k=k-1, \textbf{(27)})\, x + subs(k=k-2, \textbf{(27)})$

$$\frac{1}{2}\left(\frac{1}{x+\sqrt{x^2-1}}\right)^k+\frac{1}{2}\left(-\frac{1}{-x+\sqrt{x^2-1}}\right)^k-2\left(\frac{1}{2}\left(\frac{1}{x+\sqrt{x^2-1}}\right)^{(k-1)}+\frac{1}{2}\left(-\frac{1}{-x+\sqrt{x^2-1}}\right)^{(k-1)}\right)x+\frac{1}{2}\left(\frac{1}{x+\sqrt{x^2-1}}\right)^{(k-2)}+\frac{1}{2}\left(-\frac{1}{-x+\sqrt{x^2-1}}\right)^{(k-2)} \qquad \textbf{(31)}$$

> $normal(\%, expanded)$

$$0 \qquad \textbf{(32)}$$

>

Since the result is obvious further explanation is needless

What Have You Learnt About Maple?.

- The printf procedure is one of the worst constructions that mankind has ever created concerning the programming languages. This procedure is used for achieving formatted printing We will not go into details regarding its whole syntax. We would rather try to explain the

```
printf(`%s%u%4u  %a\n`,'`n=`', i, lcoeff(t[i],x), t[i])
```

command used in the worksheet. The printf command above executes the formatted display of four objects, that is, of the n= string, the i and the: `lcoeff(t[i], x)` expressions which are evaluated at a positive integer and of the `t[i]` which is evaluated at an algebraic expression `t[i]`

The first parameter of the printf is a string which contains the format specifications. All format specifications begin with a % sign. The system makes the objects to be displayed and the format specifications correspond with each other from left to right. According to this, the specification of the n= string is %s which means that the object to be represented is a string. The specification of the i expression is %d which indicates that the i should be displayed as an integer with a sign.

The 4 appearing in the %4dspecification of the `lcoeff(t[i], x)` expression is the width of the field in which the value of the expression is displayed. The %a specification of the t[i] shows that the t[i] should be displayed as a Maple expression. Finally, the \n requires linefeeds following the display of the four objects.

•The simplify procedure is another general simplification mechanism of Maple besides the normal procedure.

•The normal procedure creates the numerator and denominator of the rational function in an expanded syntax with the help of the expand option.

•The rsolve procedure is used to solve recurrent relations

Exercises

1. Prove that the Chebyshev polynomials fulfil the following differential equation: $(1-x^2).y''(x) - x.y'(x) + n^2.y(x)$. Is there another solution for the differential equation besides the Chebyshev polynomials?

2. Prove $\sum_{i=0}^{\infty} t_i = \frac{1}{2}$ in which case the ti is the ith Chebyshev polynomial.

3. Show the orthogonal properties of the Chebyshev polynomials in the [-1,1] interval.

If $n \neq m$ then $\int_{-1}^{1} \frac{t_n . t_m}{\sqrt{1-x^2}} \, dx = 0.$

If $n \neq 0$ then $\int_{-1}^{1} \frac{t_n^2}{\sqrt{1-x^2}} \, dx = \frac{\pi}{2}$

4. Prove the Rodriques' formula $t_n = -2^n . n! . \sqrt{1-x^2} . \frac{\frac{d^n}{dx^n}\left(\left(1-x^2\right)^{\left(n-\frac{1}{2}\right)}\right)}{(2^n)!}$ in which case the tn is the nth Chebyshev polynomial.